

【企画セッション】欠測のあるデータの解析のためのSASプログラムの紹介
～データ発生・DIAマクロとプロシジャの進展～

(6) MIプロシジャで実行可能なPattern Mixture ModelとMultiple Imputationに基づく解析

渡邊 大丞^{1,2}

(¹日本製薬工業協会 医薬品評価委員会 データサイエンス部会 タスクフォース4
欠測のあるデータの解析検討チーム, ²サノフィ株式会社)

【Planning session】Introduction of the SAS programs to analyze the missing data

(6) Analysis of Pattern Mixture Model and Multiple Imputation using the MI procedure

Daisuke Watanabe^{1,2}

(¹ The team for statistical analysis of data analysis with missing data, task force 4,
data science expert committee, drug evaluation committee,
Japan Pharmaceutical Manufacturers Association,
² Sanofi K.K.)

要旨:

- SAS V9.4より、MIプロシジャで使用可能になったMNARステートメントを使用する際の留意点を提示する。
- 本発表は応答変数が単調な欠測の場合で示す。
- うつ病の第III相試験を想定したシミュレーションデータを使用した解析結果を提示する。

キーワード: Monotone, MNAR, PMM, MI, CCMV, NCMV, Controlled imputation, Delta-adjustment method

発表構成

- MIプロシジャ(MNARステートメント)
 - ① 制約条件を用いたPMM(CCMV, NCMV)
 - ② Controlled imputation
 - ③ Delta-adjustment method
- シミュレーションデータの解析

本発表で紹介する統計手法の理論については以下を参照

- JPMALレポート(2016)
- JPMAシンポジウム(高橋、2015)
- 計量生物セミナー(土居、高橋、2015)

MI プロシジャ

Syntax: MI Procedure

The following statements are available in the MI procedure:

PROC MI *< options >* ;

BY *variables* ;

CLASS *variables* ;

EM *< options >* ;

FCS *< options >* ;

FREQ *variable* ;

MCMC *< options >* ;

MNAR *options* ;

MONOTONE *< options >* ;

TRANSFORM *transform (variables< / options >) < ... transform (variables< / options >) >* ;

VAR *variables* ;

← MNARを仮定した補完

SAS/STAT® 14.1 User's Guide The MI Procedure

MNARステートメント

```
PROC MI DATA=data-set SEED=number NIMPUTE=number OUT=data-set,
  CLASS variables;
  MONOTONE method (imputed <=effects>/ options);
  MNAR MODEL ( imputed-variables / model-options)
    ADJUST ( imputed-variable / adjust-options);
  VAR variables;
RUN;
```

記号の説明:

Y0: ベースライン値の観測値(必ず測定されている)

Y1,Y2,Y3: ベースライン後の観測値、TRT: 投与群

	Y0	Y1	Y2	Y3
パターンA	○	○	○	○
パターンB	○	○	○	×
パターンC	○	○	×	×
パターンD	○	×	×	×

TRT	Y0	Y1	Y2	Y3
1	21.0	20.0	21.5	23.0
1	21.0	21.5	24.0	25.5
1	20.5	24.0	24.5	26.0
1	23.5	24.5	25.0	.
1	21.5	23.0	22.5	.
1	20.0	21.0	21.0	.
1	21.5	22.5	.	.
1	23.0	23.0	.	.
1	20.0	21.0	.	.
1	16.5	.	.	.
1	24.5	.	.	.
2	26.0	25.0	29.0	31.0
...
2	23.0	.	.	.
2	22.0	.	.	.



① 制約条件を用いたPMM (CCMV, NCVN)

MODELオプション 概要

- MODELオプション: 補完変数及び補完方法の指定
 - MODEL (*imputed-variables / model-options*)
 - ✓ MODEL OBS = CCMV <K=k> : Complete Case Missing Value
 - ✓ MODEL OBS = NCMV <K=k> : Neighboring Case Missing Value
 - ✓ MODEL OBS = (*obs-variable=character-list*) : 特定のCLASS変数の値を有するデータで補完を行う場合に使用
- 留意点
 - ① MONOTONEまたはFCSのいずれかのステートメントが必要である。
 - ✓ FCSはCCMV及びNCMVは使用不可能。MONOTONEは全て使用可能。
 - ② MODELオプションは複数入力でき、それぞれ複数の補完変数を指定可能。同一補完変数を複数のMODELオプションに設定できない。
 - ③ 補完変数に指定されていない、欠測のある変数はMARで補完される。

MODEL オプション 留意点①

× PROC MI DATA=*data-set* SEED=... NIMPUTE=... OUT=*data-set*,
FCS options;
MNAR MODEL(Y1/ MODELOBS= **CCMV**);
VAR Y0 Y1;
RUN;

ERROR: The MODELOBS=CCMV option in the MNAR statement is allowed only if the MONOTONE statement is also specified.

× PROC MI DATA=*data-set* SEED=... NIMPUTE=... OUT=*data-set*,
FCS options;
MNAR MODEL(Y1/ MODELOBS= **NCMV**);
VAR Y0 Y1;
RUN;

ERROR: The MODELOBS=NCMV option in the MNAR statement is allowed only if the MONOTONE statement is also specified.

MODEL オプション 留意点②

O PROC MI DATA=*data-set* SEED=... NIMPUTE=... OUT=*data-set*,
MONOTONE *method (imputed <= effects>/ options)*;

```
MNAR MODEL(Y1/ MODELOBS = CCMV)  
        MODEL(Y2/ MODELOBS = CCMV);
```

```
VAR Y0 Y1 Y2;
```

```
RUN;
```

- ① MODEL(Y1 Y2/ MODELOBS= CCMV)でも良い
- ② 各変数(各時点ごと)に異なるModel optionが指定可能

X PROC MI DATA=*data-set* SEED=... NIMPUTE=... OUT=*data-set*,
MONOTONE *method (imputed <= effects>/ options)*;

```
MNAR MODEL(Y1 / MODELOBS = CCMV)
```

```
        MODEL(Y1 / MODELOBS = NCMV);
```

```
VAR Y0 Y1 Y2;
```

```
RUN;
```

ERROR: No more than one MODEL option is allowed in the MNAR statement for each imputed variable.

Y2にすると異なる補完ができる

MODELオプション 留意点③

```
PROC MI DATA=data-set SEED=... NIMPUTE=... OUT=data-set,  
MONOTONE method (imputed <=effects>/ options);
```

```
MNAR MODEL(Y1 Y2 / model-options);
```

```
VAR Y0 Y1 Y2;
```

```
RUN;
```

Y1, Y2ともにModel optionに基づき補完される

```
PROC MI DATA=data-set SEED=... NIMPUTE=... OUT=data-set,  
MONOTONE method (imputed <=effects>/ options);
```

```
MNAR MODEL( Y2 / model-options);
```

```
VAR Y0 Y1 Y2;
```

```
RUN;
```

Y2はModel optionに基づき補完、Y1はMARに
基づき補完される

MODELオプション CCMV: Kの数字の意味

CCMV: Complete Case Missing Value

- 4時点で観測を行い、その観測値を Y_0 (ベースライン), Y_1, Y_2, Y_3 とする。
- 単調な欠測を仮定し、 Y_0 は必ず観測されると仮定する。
- CCMV(K=k)は、以下の症例の情報から欠測を補完する。
 - K=1 (default): Y_3 が観測されている症例(完了例)
 - K=2: Y_2 が観測されている症例
 - K \geq 3: Y_1 が観測されている症例【MAR(ACMV)と同じ結果】

ACMV: Available Case Missing Value

	Y_0	Y_1	Y_2	Y_3
パターンA	○	○	○	○
パターンB	○	○	○	×
パターンC	○	○	×	×
パターンD	○	×	×	×

MODELオプション NCMV: Kの数字の意味

NCMV: Neighboring Case Missing Value

- 4時点で観測を行い、その観測値を Y_0 (ベースライン), Y_1, Y_2, Y_3 とする。
- 単調な欠測を仮定し、 Y_0 は必ず観測されると仮定する。
- NCMV ($K=k$)は、以下の症例の情報から欠測を補完する。
 - $K=1$ (default): 1番目に近い欠測パターン (Y_1 の欠測をパターンCで補完)
 - $K=2$: 1~2番目に近い欠測パターン (Y_1 の欠測をパターンB・Cで補完)
 - $K \geq 3$: 1~3番目に近い欠測パターン【MAR (ACMV)と同じ結果】

	Y_0	Y_1	Y_2	Y_3
パターンA	○	○	○	○
パターンB	○	○	○	×
パターンC	○	○	×	×
パターンD	○	×	×	×

Complete Case
に基づく補完

MODELオプション MODEL OBS = CCMV [CCMV(K=1)]

```
PROC MI DATA=data-set SEED=... NIMPUTE=... OUT=data-set;
  CLASS TRT; MONOTONE REG (/DETAILS);
  MNAR MODEL(Y1 Y2 Y3/ MODEL OBS = CCMV);
  VAR TRT Y0 Y1 Y2 Y3;
```

RUN; Observations Used for Imputation Models Under MNAR Assumption

```
Imputed
Variable      Observations

Y1            Complete Cases ←パターンA
Y2            Complete Cases ←パターンA
Y3            Complete Cases ←パターンA
```

	Y0	Y1	Y2	Y3
パターンA	○	○	○	○
パターンB	○	○	○	×
パターンC	○	○	×	×
パターンD	○	×	×	×

← Complete Case

TRT	Y0	Y1	Y2	Y3
1	21.0	20.0	21.5	23.0
1	21.0	21.5	24.0	25.5
1	20.5	24.0	24.5	26.0
1	23.5	24.5	25.0	25.8
1	21.5	23.0	22.5	24.9
1	20.0	21.0	21.0	25.5
1	21.5	22.5	22.2	22.8
1	23.0	23.0	23.5	25.1
1	20.0	21.0	20.8	22.4
1	16.5	19.6	17.8	19.2
1	24.5	23.7	24.7	24.3
2	26.0	25.0	29.0	31.0
...
2	23.0	23.8	26.1	30.3
2	22.0	23.8	25.2	26.4

MODEL オプション MODEL OBS = CCMV (K=2)

```
PROC MI DATA=data-set SEED=... NIMPUTE=... OUT=data-set;
  CLASS TRT; MONOTONE REG (/DETAILS);
  MNAR MODEL(Y1 Y2 Y3/ MODEL OBS = CCMV (K=2));
  VAR TRT Y0 Y1 Y2 Y3;
```

RUN; Observations Used for Imputation Models Under MNAR Assumption

Imputed			
Variable	Observations		
Y1	Nonmissing TRT, ..., Y2	←パターンA+B	
Y2	Nonmissing TRT, ..., Y2	←パターンA+B	
Y3	Complete Cases	←パターンA	

	Y0	Y1	Y2	Y3
パターンA	○	○	○	○
パターンB	○	○	○	×
パターンC	○	○	×	×
パターンD	○	×	×	×

← Complete Case

TRT	Y0	Y1	Y2	Y3
1	21.0	20.0	21.5	23.0
1	21.0	21.5	24.0	25.5
1	20.5	24.0	24.5	26.0
1	23.5	24.5	25.0	25.8
1	21.5	23.0	22.5	24.9
1	20.0	21.0	21.0	25.5
1	21.5	22.5	20.3	21.2
1	23.0	23.0	22.0	23.8
1	20.0	21.0	19.1	20.9
1	16.5	18.6	15.3	17.3
1	24.5	24.5	23.3	23.0
2	26.0	25.0	29.0	31.0
...
2	23.0	23.5	27.3	31.4
2	22.0	23.2	25.9	27.1

MODEL オプション MODEL OBS = CCMV (K=3)

```
PROC MI DATA=data-set SEED=... NIMPUTE=... OUT=data-set,
  CLASS TRT; MONOTONE REG (/DETAILS);
  MNAR MODEL(Y1 Y2 Y3/ MODEL OBS = CCMV (K=3));
  VAR TRT Y0 Y1 Y2 Y3;
```

欠測がある変数の個数以上はk=3と同じ結果

RUN; Observations Used for Imputation Models Under MNAR Assumption

```
Imputed
Variable      Observations

Y1            Nonmissing TRT, ..., Y1 ←パターンA+B+C
Y2            Nonmissing TRT, ..., Y2 ←パターンA+B
Y3            Complete Cases        ←パターンA
```

	Y0	Y1	Y2	Y3
パターンA	○	○	○	○
パターンB	○	○	○	×
パターンC	○	○	×	×
パターンD	○	×	×	×

← Complete Case

MAR (ACMV) に基づく
補完と同じ結果になる

TRT	Y0	Y1	Y2	Y3
1	21.0	20.0	21.5	23.0
1	21.0	21.5	24.0	25.5
1	20.5	24.0	24.5	26.0
1	23.5	24.5	25.0	25.8
1	21.5	23.0	22.5	24.9
1	20.0	21.0	21.0	25.5
1	21.5	22.5	20.3	21.2
1	23.0	23.0	22.0	23.8
1	20.0	21.0	19.1	20.9
1	16.5	20.0	15.4	17.1
1	24.5	23.2	23.2	23.2
2	26.0	25.0	29.0	31.0
...
2	23.0	24.0	27.3	31.3
2	22.0	24.1	26.0	27.0

一番近い欠測
パターンで補完

MODELオプション MODEL OBS = NCMV [NCMV(K=1)]

```
PROC MI DATA=data-set SEED=... NIMPUTE=... OUT=data-set,
  CLASS TRT; MONOTONE REG (/DETAILS);
  MNAR MODEL(Y1 Y2 Y3/ MODEL OBS = NCMV);
  VAR TRT Y0 Y1 Y2 Y3;
```

RUN; Observations Used for Imputation Models Under MNAR Assumption

Imputed
Variable Observations

Y1 Nonmissing TRT, ..., Y1; Missing Y2, Y3 ← **パターンC**
 Y2 Nonmissing TRT, ..., Y2; Missing Y3 ← **パターンB**
 Y3 Complete Cases ← **パターンA**

	Y0	Y1	Y2	Y3
パターンA	○	○	○	○
パターンB	○	○	○	×
パターンC	○	○	×	×
パターンD	○	×	×	×

TRT	Y0	Y1	Y2	Y3
1	21.0	20.0	21.5	23.0
1	21.0	21.5	24.0	25.5
1	20.5	24.0	24.5	26.0
1	23.5	24.5	25.0	26.2
1	21.5	23.0	22.5	24.2
1	20.0	21.0	21.0	23.9
1	21.5	22.5	24.7	25.4
1	23.0	23.0	23.4	23.6
1	20.0	21.0	16.2	21.2
1	16.5	30.2	-16.2	14.5
1	24.5	21.8	32.8	26.6
2	26.0	25.0	29.0	31.0
...
2	23.0	21.3	26.2	27.2
2	22.0	23.7	18.3	26.2

注Kが2以上の場合はBack-upを参照



② Controlled imputation

MODELオプション

MODEL OBS=(*obs*-*variables*=*character-list*)

PROC MI DATA=*data-set* SEED=... NIMPUTE=... OUT=*data-set*,

CLASS TRT; MONOTONE REG (/DETAILS);

MNAR MODEL(Y1 Y2 Y3/ MODEL OBS = (**TRT='2'**));

VAR Y0 Y1 Y2 Y3;

RUN;

必ず quotation mark で囲む

- ① TRT=2のデータで補完モデルを作る
- ② 両群の欠測データを補完

Observations Used for Imputation Models Under MNAR Assumption

Imputed
Variable Observations

Y1 TRT = 2 ← TRT=2のデータ
Y2 TRT = 2 ← TRT=2のデータ
Y3 TRT = 2 ← TRT=2のデータ

	TRT	Y0	Y1	Y2	Y3
パターンA	1	○	○	○	○
パターンB	1	○	○	○	×
パターンC	1	○	○	×	×
パターンD	1	○	×	×	×

	TRT	Y0	Y1	Y2	Y3
パターンA	2	○	○	○	○
パターンB	2	○	○	○	×
パターンC	2	○	○	×	×
パターンD	2	○	×	×	×

TRT	Y0	Y1	Y2	Y3
1	21.0	20.0	21.5	23.0
1	21.0	21.5	24.0	25.5
1	20.5	24.0	24.5	26.0
1	23.5	24.5	25.0	26.6
1	21.5	23.0	22.5	24.5
1	20.0	21.0	21.0	21.9
1	21.5	22.5	25.4	26.7
1	23.0	23.0	22.6	24.7
1	20.0	21.0	19.8	19.8
1	16.5	22.4	18.9	21.0
1	24.5	23.8	24.4	27.9
2	26.0	25.0	29.0	31.0
...
2	23.0	24.9	24.7	24.7
2	22.0	23.8	25.2	27.9

伊藤, 西本 (2014)



③ Delta-adjustment method


ADJUSTオプション 概要 (1/4)

- ADJUST option: 補完データの調整方法の指定
 - ADJUST(*imputed-variable* / *adjust-options*)
 - ✓ ADJUST(*imputed-variable* (EVENT='level') / *adjust-options*)
 - MONOTONEステートメント、FCSステートメントともに使用可能
 - Adjust-options
 - ✓ ADJUSTOBS = (*obs-variable=character-list*): 調整する集団の特定
 - ✓ SCALE = *c*: 補完変数の補完値を調整するScale parameter
 - ✓ SHIFT | DELTA = δ : 補完変数の補完値を調整するShift parameter
 - ✓ SIGMA = σ : 補完変数の補完値を調整するShift parameterの標準偏差
- \tilde{Y} : 補完値、 \tilde{Y}^* : 調整値
 - $\tilde{Y}^* = \underbrace{c}_{\substack{\uparrow \\ \text{SCALE}=c}} \tilde{Y} + \delta^*、\delta^* \sim N(\underbrace{\delta}_{\substack{\uparrow \\ \text{SHIFT|DELTA}=\delta}}, \underbrace{\sigma^2}_{\substack{\uparrow \\ \text{SIGMA}=\sigma}})$


ADJUSTオプション

概要 (2/4)


MARで補完した Y_2 の補完値を調整: $\tilde{Y}_2^* = c\tilde{Y}_2 + \delta^*, \delta^* \sim N(0, \sigma^2)$



SCALE=c



SHIFT= δ




SIGMA= σ

●—● 観測値 ○·····○ MARで補完した値 ○---○ 最終的な補完値

最終的な補完値は以降の時点の補完に使用される

ADJUSTオプション 概要 (3/4)

MARで補完した各時点の補完値を調整 (SHIFT= δ) した場合



●—● 観測値

○-...-○ MARで補完した値

○- - - - - ○ Y_2 を調整した時の最終的な補完値

○- - ○ - - - ○ Y_2, Y_3 を調整した時の最終的な補完値

○-○-○-○ Y_2, Y_3, Y_4 を調整した時の最終的な補完値

ADJUSTオプション 概要(4/4)

- 留意点

- ① ADJUSTOBSを指定しない場合、全ての補完値を調整する。
- ② ADJUSTオプションは複数個設定可能。しかし各々1つの補完変数しか指定できない。
 - ある欠測パターンを示す変数をCLASS変数とADJUSTOBSに入れることで欠測パターンごとに調整が可能。
- ③ SCALEの値は正の値。連続量の補完値の調整のみ使用可。SCALE又はSHIFT|DELTAは、SAS datasetを使用して各補完番号(_IMPUTATION_)に対して異なる値を指定することができる。
 - ✓ PARMS(parms-options)=SAS-dataset
 - SHIFT | DELTA=variable
 - SCALE = variable
 - ✓ SAS datasetに、補完番号を示す”_IMPUTATION_”を入れる必要がある。
 - ✓ SIGMAと併用することはできる。

ADJUSTオプション 留意点②



```
PROC MI DATA=data-set SEED=... NIMPUTE=... OUT=data-set,  
  MNAR ADJUST (Y1 Y2 / adjust-options);  
  VAR Y0 Y1 Y2;
```

RUN;

ERROR 22-322: Syntax error, expecting one of the following: (, /.
ERROR 202-322: The option or parameter is not recognized and will be ignored.



```
PROC MI DATA=data-set SEED=... NIMPUTE=... OUT=data-set,  
  MNAR ADJUST (Y1 / adjust-options)  
    ADJUST (Y2 / adjust-options);  
  VAR Y0 Y1 Y2;
```

RUN;

ADJUSTオプション 留意点③

❌ `PROC MI DATA=data-set SEED=... NIMPUTE=... OUT=data-set,
MNAR ADJUST (Y1/ SCALE=-c);
VAR Y0 Y1;
RUN;`

ERROR: The parameter value -c is less than the minimum allowed value of 0.

❌ `PROC MI DATA=data-set SEED=... NIMPUTE=... OUT=data-set,
MNAR ADJUST (imputed-variables (EVENT='level') / SCALE=c);
VAR variables;
RUN;`

ERROR: The SCALE= option in the ADJUST option in the MNAR statement is not allowed for the logistic regression imputed variable XXXX.

ADJUSTオプション SCALE = c

```
PROC MI DATA=data-set SEED=... NIMPUTE=... OUT=data-set,
  CLASS TRT; MONOTONE REG (/DETAILS);
  MNAR ADJUST(Y3/ SCALE= 2); ○ ○ ○
  VAR TRT Y0 Y1 Y2 Y3;
```

この場合Y1とY2は
MAR(ACMV)に
基づく補完

RUN;

MNAR Adjustments to Imputed Values

Imputed Variable	Scale
Y3	2.0000

	Y0	Y1	Y2	Y3
パターンA	○	○	○	○
パターンB	○	○	○	$2 \times \tilde{y}$
パターンC	○	○	\tilde{y}	$2 \times \tilde{y}$
パターンD	○	\tilde{y}	\tilde{y}	$2 \times \tilde{y}$

\tilde{y} : MARに基づく補完値

Before					After				
TRT	Y0	Y1	Y2	Y3	TRT	Y0	Y1	Y2	Y3
1	21.0	20.0	21.5	23.0	1	21.0	20.0	21.5	23.0
1	21.0	21.5	24.0	25.5	1	21.0	21.5	24.0	25.5
1	20.5	24.0	24.5	26.0	1	20.5	24.0	24.5	26.0
1	23.5	24.5	25.0	25.8	1	23.5	24.5	25.0	51.5
1	21.5	23.0	22.5	24.9	1	21.5	23.0	22.5	49.8
1	20.0	21.0	21.0	25.5	1	20.0	21.0	21.0	50.9
1	21.5	22.5	20.3	21.2	1	21.5	22.5	20.3	42.4
1	23.0	23.0	22.0	23.8	1	23.0	23.0	22.0	47.5
1	20.0	21.0	19.1	21.0	1	20.0	21.0	19.1	42.0
1	16.5	20.0	15.4	17.0	1	16.5	20.0	15.4	34.1
1	24.5	23.2	23.2	23.2	1	24.5	23.2	23.2	46.4
2	26.0	25.0	29.0	31.0	2	26.0	25.0	29.0	31.0
...
2	23.0	24.0	27.3	31.3	2	23.0	24.0	27.3	62.7
2	22.0	24.1	26.0	26.9	2	22.0	24.1	26.0	53.8

ADJUSTオプション SHIFT | DELTA = δ

```
PROC MI DATA=data-set SEED=... NIMPUTE=... OUT=data-set,
  CLASS TRT; MONOTONE REG (/DETAILS);
  MNAR ADJUST(Y3/ SHIFT = 2);
  VAR TRT Y0 Y1 Y2 Y3;
```

この場合Y1とY2は
MAR(ACMV)に
基づく補完

RUN;

MNAR Adjustments to Imputed Values

Imputed Variable	Shift
Y3	2.0000

	Y0	Y1	Y2	Y3
パターンA	○	○	○	○
パターンB	○	○	○	$\tilde{y} + 2$
パターンC	○	○	\tilde{y}	$\tilde{y} + 2$
パターンD	○	\tilde{y}	\tilde{y}	$\tilde{y} + 2$

\tilde{y} : MARに基づく補完値

Before					After				
TRT	Y0	Y1	Y2	Y3	TRT	Y0	Y1	Y2	Y3
1	21.0	20.0	21.5	23.0	1	21.0	20.0	21.5	23.0
1	21.0	21.5	24.0	25.5	1	21.0	21.5	24.0	25.5
1	20.5	24.0	24.5	26.0	1	20.5	24.0	24.5	26.0
1	23.5	24.5	25.0	25.8	1	23.5	24.5	25.0	27.8
1	21.5	23.0	22.5	24.9	1	21.5	23.0	22.5	26.9
1	20.0	21.0	21.0	25.5	1	20.0	21.0	21.0	27.5
1	21.5	22.5	20.3	21.2	1	21.5	22.5	20.3	23.2
1	23.0	23.0	22.0	23.8	1	23.0	23.0	22.0	25.8
1	20.0	21.0	19.1	21.0	1	20.0	21.0	19.1	23.0
1	16.5	20.0	15.4	17.0	1	16.5	20.0	15.4	19.0
1	24.5	23.2	23.2	23.2	1	24.5	23.2	23.2	25.2
2	26.0	25.0	29.0	31.0	2	26.0	25.0	29.0	31.0
...
2	23.0	24.0	27.3	31.3	2	23.0	24.0	27.3	33.3
2	22.0	24.1	26.0	26.9	2	22.0	24.1	26.0	28.9

ADJUSTオプション

$$\text{SIGMA} = \sigma$$

```
PROC MI DATA=data-set SEED=... NIMPUTE=... OUT=data-set,
```

```
CLASS TRT; MONOTONE REG (/DETAILS);
```

```
MNAR ADJUST(Y3/ SHIFT=2 SIGMA = 1);
```

```
VAR TRT Y0 Y1 Y2 Y3;
```

```
RUN;
```

正規乱数を発生させ、
その値をSHIFTパラ
メータとして調整

MNAR Adjustments to Imputed Values

```
Imputed
Variable      Imputation      Shift
Y3              1          1.4031
```

	Y0	Y1	Y2	Y3
パターンA	○	○	○	○
パターンB	○	○	○	$\tilde{y} + \delta^*$
パターンC	○	○	\tilde{y}	$\tilde{y} + \delta^*$
パターンD	○	\tilde{y}	\tilde{y}	$\tilde{y} + \delta^*$

\tilde{y} : MARに基づく補完値、 $\delta^* \sim N(2,1)$

Before

TRT	Y0	Y1	Y2	Y3
1	21.0	20.0	21.5	23.0
1	21.0	21.5	24.0	25.5
1	20.5	24.0	24.5	26.0
1	23.5	24.5	25.0	27.5
1	21.5	23.0	22.5	26.9
1	20.0	21.0	21.0	23.5
1	21.5	22.5	21.2	23.7
1	23.0	23.0	23.3	24.8
1	20.0	21.0	21.2	24.9
1	16.5	19.9	15.0	17.6
1	24.5	23.9	26.6	28.6
2	26.0	25.0	29.0	31.0
...
2	23.0	24.0	25.8	26.2
2	22.0	21.2	22.4	24.9

After

TRT	Y0	Y1	Y2	Y3
1	21.0	20.0	21.5	23.0
1	21.0	21.5	24.0	25.5
1	20.5	24.0	24.5	26.0
1	23.5	24.5	25.0	28.9
1	21.5	23.0	22.5	28.3
1	20.0	21.0	21.0	24.9
1	21.5	22.5	21.2	25.1
1	23.0	23.0	23.3	26.2
1	20.0	21.0	21.2	26.3
1	16.5	19.9	15.0	19.0
1	24.5	23.9	26.6	30.0
2	26.0	25.0	29.0	31.0
...
2	23.0	24.0	25.8	27.6
2	22.0	21.2	22.4	26.3

ADJUSTオプション

SHIFT | DELTA = δ , ADJUSTOBS

```
PROC MI DATA=data-set SEED=... NIMPUTE=... OUT=data-set,
  CLASS TRT; MONOTONE REG (/DETAILS);
  MNAR ADJUST(Y3/ SHIFT = 2 ADJUSTOBS=(TRT='1'));
  VAR TRT Y0 Y1 Y2 Y3;
RUN;
```

TRT=1のY3以外
はMAR(ACMV)に
基づく補完

MNAR Adjustments to Imputed Values

Imputed Variable	Observations	Shift
Y3	TRT = 1	2.0000

Before					After				
TRT	Y0	Y1	Y2	Y3	TRT	Y0	Y1	Y2	Y3
1	21.0	20.0	21.5	23.0	1	21.0	20.0	21.5	23.0
1	21.0	21.5	24.0	25.5	1	21.0	21.5	24.0	25.5
1	20.5	24.0	24.5	26.0	1	20.5	24.0	24.5	26.0
1	23.5	24.5	25.0	25.8	1	23.5	24.5	25.0	27.8
1	21.5	23.0	22.5	24.9	1	21.5	23.0	22.5	26.9
1	20.0	21.0	21.0	25.5	1	20.0	21.0	21.0	27.5
1	21.5	22.5	20.3	21.2	1	21.5	22.5	20.3	23.2
1	23.0	23.0	22.0	23.8	1	23.0	23.0	22.0	25.8
1	20.0	21.0	19.1	21.0	1	20.0	21.0	19.1	23.0
1	16.5	20.0	15.4	17.0	1	16.5	20.0	15.4	19.0
1	24.5	23.2	23.2	23.2	1	24.5	23.2	23.2	25.2
2	26.0	25.0	29.0	31.0	2	26.0	25.0	29.0	31.0
...
2	23.0	24.0	27.3	31.3	2	23.0	24.0	27.3	31.3
2	22.0	24.1	26.0	26.9	2	22.0	24.1	26.0	26.9

ADJUSTオプション SHIFT | DELTA = δ , PARMS

```
PROC MI DATA=data-set SEED=... NIMPUTE=... OUT=data-set,
  CLASS TRT; MONOTONE REG (/DETAILS);
  MNAR ADJUST(Y3/ ADJUSTOBS=(Trt='1') parms(shift=shift1)=data-set)
  ADJUST(Y3/ ADJUSTOBS=(Trt='2') parms(shift=shift2)=data-set);
  VAR TRT Y0 Y1 Y2 Y3;
  RUN;
```

群ごとに異なる
値でY3を調整

	IMPUTATION	SHIFT1	SHIFT2
1	1	2	4

MNAR Adjustments to Imputed Values

Imputed Variable	Imputation	Observations	Shift
Y3	1	TRT = 1	2.0000
	1	TRT = 2	4.0000


Before					After				
TRT	Y0	Y1	Y2	Y3	TRT	Y0	Y1	Y2	Y3
1	21.0	20.0	21.5	23.0	1	21.0	20.0	21.5	23.0
1	21.0	21.5	24.0	25.5	1	21.0	21.5	24.0	25.5
1	20.5	24.0	24.5	26.0	1	20.5	24.0	24.5	26.0
1	23.5	24.5	25.0	25.8	1	23.5	24.5	25.0	27.8
1	21.5	23.0	22.5	24.9	1	21.5	23.0	22.5	26.9
1	20.0	21.0	21.0	25.5	1	20.0	21.0	21.0	27.5
1	21.5	22.5	20.3	21.2	1	21.5	22.5	20.3	23.2
1	23.0	23.0	22.0	23.8	1	23.0	23.0	22.0	25.8
1	20.0	21.0	19.1	21.0	1	20.0	21.0	19.1	23.0
1	16.5	20.0	15.4	17.0	1	16.5	20.0	15.4	19.0
1	24.5	23.2	23.2	23.2	1	24.5	23.2	23.2	25.2
2	26.0	25.0	29.0	31.0	2	26.0	25.0	29.0	31.0
...
2	23.0	24.0	27.3	31.3	2	23.0	24.0	27.3	35.3
2	22.0	23.8	25.2	26.3	2	22.0	24.1	26.0	30.9



シミュレーションデータの解析

シミュレーションデータの解析

◎うつ病の第III相試験を想定したシミュレーションデータ



● 主要評価項目: HAM-D
→ スコア低下: 改善

● 実薬群 vs プラセボ群
・1群120例(ベースライン)

	ベースライン		時点1		時点2		時点3		時点4	
	例数	平均 (SD)	例数	平均 (SD)	例数	平均 (SD)	例数	平均 (SD)	例数	平均 (SD)
実薬群	120	20.0 (4.1)	100	14.9 (4.4)	92	12.8 (6.5)	86	10.1 (7.5)	83	8.1 (8.7)
プラセボ群	120	20.2 (3.9)	106	15.9 (4.6)	100	13.3 (5.3)	94	12.1 (7.2)	88	11.3 (8.2)

シミュレーションデータの解析 解析内容(1/2)

- ① LOCF ANCOVA
- ② Complete case analysis (ANCOVA)
- ③ MMRM
- ④ MI ANCOVA、MI MMRM
- ⑤ Placebo multiple imputation (pMI MMRM)
- ⑥ Pattern-mixture model (CCMV MMRM, NCMV MMRM)
- ⑦ Delta-adjustment method(実薬群の全時点を調整) ($\delta=0, 0.2, 0.4, 0.6, 0.8$) (MMRM)
- ⑧ Delta-adjustment method(実薬群の中止後最初の時点を調整) ($\delta=0, 0.2, 0.4, 0.6, 0.8, 1.0, 1.2, 1.4, 1.6, 1.8, 2.0$) (MMRM)

補完回数: 50回

シミュレーションデータの解析 解析内容(2/2)

	欠測値の補完	解析方法	使用した プロシジャ	MIプロシジャの MODELオプション	MIプロシジャの ADJUSTオプション
①	LOCF	ANCOVA	MIXED	使用せず	使用せず
②	なし	ANCOVA	MIXED	使用せず	使用せず
③	なし	MMRM	MIXED	使用せず	使用せず
④	MI(Regression)	ANCOVA, MMRM	MI, MIXED, MIANALYZE	使用せず	使用せず
⑤	プラセボ群データを用いた補完モデル	MMRM	MI, MIXED, MIANALYZE	MODELOBS= (TRT='プラセボ群')	使用せず
⑥	CCMV, NCMV	MMRM	MI, MIXED, MIANALYZE	MODELOBS=CCMV, MODELOBS=NCMV	使用せず
⑦ ⑧	MI(Regression)の補完値を調整値(δ)	MMRM	MI, MIXED, MIANALYZE	使用せず	SHIFT= δ ADJUSTOBS=(TRT='調整を行う群')

シミュレーションデータの解析 解析結果(1/6)

	解析手法	群間差	群間差のSE	95% CI	P-value
①	LOCF ANCOVA	-2.604	1.090	-4.753 to -0.455	0.0178
②	Complete case (ANCOVA)	-3.206	1.223	-5.620 to -0.793	0.0095
③	MMRM	-2.873	1.183	-5.207 to -0.538	0.0162
④	MI ANCOVA	-2.856	1.182	-5.175 to -0.536	0.0159
④	MI MMRM	-2.845	1.182	-5.165 to -0.524	0.0163
⑤	pMI MMRM	-2.048	1.166	-4.335 to 0.239	0.0792
⑥	CCMV MMRM	-2.920	1.171	-5.217 to -0.623	0.0128
⑥	NCMV MMRM	-4.009	2.436	-8.848 to 0.830	0.1033


シミュレーションデータの解析 解析結果(2/6)

	解析手法	群間差	群間差のSE	95% CI	P-value
⑦	Delta-adjustment method [全時点]				
	$\delta=0$	-2.845	1.182	-5.165 to -0.524	0.0163
	$\delta=0.2$	-2.676	1.184	-5.001 to -0.352	0.0241
	$\delta=0.4$	-2.508	1.187	-4.837 to -0.179	0.0348
	$\delta=0.6$	-2.340	1.190	-4.675 to -0.005	0.0495
	$\delta=0.8$	-2.172	1.193	-4.513 to 0.170	0.0691


シミュレーションデータの解析 解析結果 (3/6)

	解析手法	群間差	群間差のSE	95% CI	P-value
⑧	Delta-adjustment method [最初の時点のみ]				
	$\delta=0$	-2.845	1.182	-5.165 to -0.524	0.0163
	$\delta=0.2$	-2.791	1.184	-5.115 to -0.468	0.0186
	$\delta=0.4$	-2.737	1.185	-5.064 to -0.411	0.0212
	$\delta=0.6$	-2.684	1.187	-5.013 to -0.354	0.0240
	$\delta=0.8$	-2.630	1.189	-4.963 to -0.297	0.0272
	$\delta=1.0$	-2.576	1.190	-4.913 to -0.240	0.0307
	$\delta=1.2$	-2.523	1.192	-4.863 to -0.183	0.0346
	$\delta=1.4$	-2.469	1.194	-4.813 to -0.125	0.0390
	$\delta=1.6$	-2.415	1.196	-4.763 to -0.068	0.0438
	$\delta=1.8$	-2.362	1.198	-4.713 to -0.010	0.0490
	$\delta=2.0$	-2.308	1.200	-4.664 to 0.048	0.0548


シミュレーションデータの解析 解析結果(4/6)



シミュレーションデータの解析 解析結果 (5/6)



シミュレーションデータの解析 解析結果(6/6)

- MMRM、MI ANCOVA及びCCMV(MMRM)の結果はほぼ同様であった。
 - LOCF ANCOVAの結果は、MMRM、MI ANCOVA及びCCMV(MMRM)の結果と大きな乖離はないが、群間差は小さかった。
 - Complete case analysisの結果は、大きな乖離はなかったが、上記の統計手法に群間差が大きかった。
- pMI及びNCMV(MMRM)の検定結果は有意ではなかった。
 - NCMVの結果は、今回の解析結果の中で最もSEが大きかった。
 - 一番近い欠測パターンの例数が影響していたと考えられる。
 - 時点4のみが欠測しているパターンが両群で9例(プラセボ:6例、実薬群:3例)
 - 時点3以降が欠測しているパターンが両群で12例(プラセボ:6例、実薬群:6例)
 - 時点2以降が欠測しているパターンが両群で14例(プラセボ:6例、実薬群:8例)
- Delta-adjustment methodの結果は、全時点を調整した場合、感度パラメータ(δ)を0.8にしたときに有意差がなくなった。また中止後最初の時点のみを調整した場合、感度パラメータ(δ)を2.0にしたときに有意差がなくなった。

まとめ

- MIプロシジャ (MNARステートメント)
 - ① 制約条件を用いたPMM (CCMV, NCMV)
 - ✓ MODELOBSオプション: CCMV(K=k)、NCMV(K=k)
 - ② Controlled imputation: MNAR MODEL(.../ **model-options**)
 - ✓ MODELOBSオプション: 補完モデルで使用する投与群を指定
 - ✓ 例: MODLOBS=(TRT="1")
 - ③ Delta-adjustment method: MNAR ADJUST(.../ **adjust-options**)
 - ✓ ADJUSTOBSオプション: 補完値を調整する投与群を指定
 - ✓ SHIFTオプション: 調整する値を指定
 - ✓ 例: SHIFT= δ ADJUSTOBS=(TRT="1")
 - 各変数 (各時点ごと) に異なる補完モデルが指定可能
 - 補完値をシフトするだけでなく、補完値に定数を掛けること (SCALE)、補完値にバラツキをもたせること (SIGMA) も可能
- シミュレーションデータの解析

参考文献

- SAS/STAT® 14.1 User's Guide The MI Procedure (2015)
- Potthoff. R. F and Roy. S.N. (1964) A Generalized Multivariate Analysis of Variance Model Useful Especially for Growth Curve Problems. *Biometrika*, 51, 313-326.
- Yuan. Y. (2014) SAS Global forum, Sensitivity Analysis in Multiple Imputation for Missing Data. SAS Global forum. Washington, DC.
- 伊藤 陽一, 西本尚樹. (2014). 多重補完法におけるPattern-Mixtureモデルに基づく感度分析. SASユーザー総会論文集.
- 土居 正明, 高橋 文博. (2015). 計量生物セミナー 臨床試験におけるEstimandと感度分析、欠測のあるデータに対する解析手法の基礎 ～ (3)MNAR を仮定した PMM の解析～
- 日本製薬工業協会 医薬品評価委員会 データサイエンス部会 2013 年度タスクフォース2/2014 年度タスクフォース4 欠測のあるデータの解析チーム (2016). 欠測のある連続量経時データに対する統計手法について
- 高橋 文博. (2015). 【日本製薬工業協会シンポジウム】臨床試験の欠測データの取り扱いに関する最近の展開と今後の課題について — 統計手法・estimandと架空の事例に対する流れの整理 — (3) PATTERN-MIXTURE MODELの解説



Back up

MODELオプション

MODELOBS = NCMV (K=k)の不具合

- 現在リリースされているバージョンでは、Kの値が1より大きく、欠測パターンのグループより小さい場合、一部の補完に完了例のデータが考慮されていない不具合が発生している。
- 例えばK=2でY2を補完する場合、本来ならば完了例(パターンA)が補完モデルに含まれるべきであるが、含まれておらずNCMV(K=1)と同じパターンBのみが含まれている補完モデルで補完されている。
- 【対応策】DATEステップで欠測パターンを特定する変数を作成し、`MODELOBS = (obs-variable=character-list)` で補完を行う。

本来行われる補完

	Y0	Y1	Y2	Y3
パターンA	○	○	○	○
パターンB	○	○	○	×
パターンC	○	○	×	×
パターンD	○	×	×	×

実際行われている補完

	Y0	Y1	Y2	Y3
パターンA	○	○	○	○
パターンB	○	○	○	×
パターンC	○	○	×	×
パターンD	○	×	×	×

MODELオプション

MODELOBS = NCMV (K=2)の対応策

```
DATA data-set2; SET data-set1;
```

```
GROUP=1; IF (Y3=.) THEN GROUP=2;
```

```
IF (Y2=.) THEN GROUP=3; IF (Y1=.) THEN GROUP=4;
```

```
IF GROUP IN (2,3) THEN Y1GROUP= 1; ELSE Y1GROUP= 0;
```

```
IF GROUP IN (1,2) THEN Y2GROUP= 1; ELSE Y2GROUP= 0;
```

} 欠測パターン

} 補完モデルに用
いる欠測パターン

```
RUN;
```

```
PROC MI DATA=data-set2 SEED=... NIMPUTE=... OUT=data-set3;
```

```
CLASS TRT GROUP Y1GROUP Y2GROUP;
```

```
MONOTONE REG (/DETAILS);
```

```
MNAR MODEL(Y1 / MODELOBS=(Y1GROUP='1'))
```

```
MODEL(Y2 / MODELOBS=(Y2GROUP='1'))
```

```
MODEL(Y3 / MODELOBS=(GROUP='1'));
```

```
VAR TRT Y0 Y1 Y2 Y3;
```

```
RUN;
```

MODELオプション

MODELOBS = NCMV (K=3)の対応策

```
DATA data-set2; SET data-set1;
```

```
GROUP=1; IF (Y3=.) THEN GROUP=2;
```

```
IF (Y2=.) THEN GROUP=3; IF (Y1=.) THEN GROUP=4;
```

```
IF GROUP IN (1,2,3) THEN Y1GROUP= 1; ELSE Y1GROUP= 0;
```

```
IF GROUP IN (1,2) THEN Y2GROUP= 1; ELSE Y2GROUP= 0;
```

} 欠測パターン

} 補完モデルに用
いる欠測パターン

```
RUN;
```

```
PROC MI DATA=data-set2 SEED=... NIMPUTE=... OUT=data-set3;
```

```
CLASS TRT GROUP Y1GROUP Y2GROUP;
```

```
MONOTONE REG (/DETAILS);
```

```
MNAR MODEL(Y1 / MODELOBS=(Y1GROUP='1'))
```

```
MODEL(Y2 / MODELOBS=(Y2GROUP='1'))
```

```
MODEL(Y3 / MODELOBS=(GROUP='1'));
```

```
VAR TRT Y0 Y1 Y2 Y3;
```

```
RUN;
```

プログラム(サンプル) - 多重補完 -

Monotone Regression Method

```
PROC MI DATA=data-set SEED=number NIMPUTE=number OUT=data-set,  
  CLASS TRT;  
  MONOTONE REG (/DETAILS);  
  VAR TRT X0 X1 X2 X3 X4;  
RUN;
```

Controlled imputation (placebo multiple imputation)

```
PROC MI DATA=data-set SEED=number NIMPUTE=number OUT=data-set,  
  CLASS TRT;  
  MONOTONE REG (/DETAILS);  
  MNAR MODEL(X1 X2 X3 X4/ MODEL OBS= (TRT='PLACEBO'));  
  VAR X0 X1 X2 X3 X4;  
RUN;
```

プログラム(サンプル) - 多重補完 -

Complete Case Missing Value (CCMV)

```
PROC MI DATA=data-set SEED=number NIMPUTE=number OUT=data-set,  
  CLASS TRT;  
  MONOTONE REG (/DETAILS);  
  MNAR MODEL(X1 X2 X3 X4/ MODELOBS= CCMV(K=k));  
  VAR TRT X0 X1 X2 X3 X4;  
RUN;
```

Neighboring Case Missing Value (NCMV)

```
PROC MI DATA=data-set SEED=number NIMPUTE=number OUT=data-set,  
  CLASS TRT;  
  MONOTONE REG (/DETAILS);  
  MNAR MODEL(X1 X2 X3 X4/ MODELOBS= NCMV(K=k));  
  VAR X0 X1 X2 X3 X4;  
RUN;
```


プログラム(サンプル) - 多重補完 -

Delta-adjustment method 【全時点】[Yuan (2014)]

```
%MACRO MIDATA( DATA=, SMIN=, SMAX=, SINC=, OUT=, SEED=, NIMP=);  
  DATA &OUT; SET _NULL_; RUN;  
  /*----- # OF SHIFT VALUES -----*/  
  %LET NCASE=%SYSEVALF( (&SMAX-&SMIN)/&SINC, CEIL );  
  /*----- IMPUTED DATA FOR EACH SHIFT -----*/  
  %DO JC=0 %TO &NCASE; %LET SJ= %SYSEVALF(&SMIN + &JC * &SINC);  
  PROC MI DATA=&DATA SEED=&SEED. NIMPUTE=&NIMP. OUT=OUTMI;  
    CLASS TRT; MONOTONE REG(/DETAILS);  
    MNAR ADJUST(X1 / SHIFT=&SJ. ADJUSTOBS=(TRT='1'))  
      ADJUST(X2 / SHIFT=&SJ. ADJUSTOBS=(TRT='1'))  
      ADJUST(X3 / SHIFT=&SJ. ADJUSTOBS=(TRT='1'))  
      ADJUST(X4 / SHIFT=&SJ. ADJUSTOBS=(TRT='1'));  
  VAR TRT X0 X1 X2 X3 X4;  
  RUN;  
  DATA OUTMI; SET OUTMI; SHIFT=&SJ; RUN; DATA &OUT; SET &OUT OUTMI; RUN;  
%END; %MEND MIDATA;
```

プログラム(サンプル) - 多重補完 -

Delta-adjustment method 【中止後最初の時点】[Yuan (2014)]

```
%MACRO MIDATA( DATA=, SMIN=, SMAX=, SINC=, OUT=, SEED=, NIMP=);  
DATA &OUT; SET _NULL_; RUN;  
/*----- # OF SHIFT VALUES -----*/  
%LET NCASE=%SYSEVALF( (&SMAX-&SMIN)/&SINC, CEIL );  
/*----- IMPUTED DATA FOR EACH SHIFT -----*/  
%DO JC=0 %TO &NCASE; %LET SJ= %SYSEVALF(&SMIN + &JC * &SINC);  
PROC MI DATA=&DATA SEED=&SEED. NIMPUTE=&NIMP. OUT=OUTMI;  
CLASS TRT GROUP; MONOTONE REG(/DETAILS);  
MNAR ADJUST(X1 / SHIFT=&SJ. ADJUSTOBS=(GROUP='5'))  
ADJUST(X2 / SHIFT=&SJ. ADJUSTOBS=(GROUP='4'))  
ADJUST(X3 / SHIFT=&SJ. ADJUSTOBS=(GROUP='3'))  
ADJUST(X4 / SHIFT=&SJ. ADJUSTOBS=(GROUP='2'));  
VAR TRT X0 X1 X2 X3 X4;  
RUN;  
DATA OUTMI; SET OUTMI; SHIFT=&SJ; RUN; DATA &OUT; SET &OUT OUTMI; RUN;  
%END; %MEND MIDATA;
```

調整する投与群の欠測パターンを示す変数を作成しておく(例ではGROUPを作成)。その変数をCLASS変数に指定する

プログラム(サンプル)-CCMV(1/2)-

/ 各症例の各時点の観測値を横に並べる(1症例1オブザベーション) */*

/ このパートを実施したい多重補完の方法に変更する */*

```
PROC MI DATA=INDS SEED=12345 NIMPUTE=50 OUT=IMPDS;
```

```
  CLASS TRT; MONOTONE REG (/DETAILS);
```

```
  MNAR MODEL(X1 X2 X3 X4/ MODELOBS= CCMV(K=1)); VAR TRT X0 X1 X2 X3 X4;
```

```
RUN;
```

/ MIXEDプロシジャ実行の為、各症例の各時点の観測値と補完値を縦に並べる */*

```
PROC SORT DATA=IMPDS; BY _IMPUTATION_ TRT ID X0; RUN;
```

```
PROC TRANSPOSE DATA=IMPDS OUT=TRANSDS PREFIX=VAL;
```

```
  BY _IMPUTATION_ TRT ID X0; VAR X1 X2 X3 X4;
```

```
RUN;
```

/ ベースラインからの変化量(DIFF)の計算及び測定時点の変数(TIME)の作成 */*

```
DATA TRANSDS;
```

```
  SET TRANSDS; BY _IMPUTATION_ TRT ID X0;
```

```
  TIME=INPUT(SUBSTR(_NAME_,2),BEST.); DIFF=VAL1-X0;
```

```
RUN;
```

プログラム(サンプル)-CCMV(2/2)-

/ 解析の実行(MMRM)*/*

```
PROC MIXED DATA = TRANSDDS(WHERE=(DIFF NE .));  
  BY _IMPUTATION_;  
  CLASS TRT TIME ID;  
  MODEL DIFF = X0 TRT TIME TRT*TIME / DDFM=KR;  
  REPEATED TIME / TYPE = UN SUBJECT = ID;  
  LSMEANS TRT*TIME / DIFF=CONTROL('2' '4') CL;  
  ODS OUTPUT DIFFS=DIFFS;  
RUN;
```



/ MIANALYZEプロシジャを用いて多重補完されたデータの解析結果を統合する*/*

```
PROC MIANALYZE PARMS=DIFFS (WHERE=(TIME EQ 4 AND _TIME EQ 4));  
  MODELEFFECTS TRT*TIME;  
RUN;
```

シミュレーションデータ


欠測パターンごとの症例数とベースラインからの変化量

欠測パターン	ベースライン	時点1	時点2	時点3	時点4	プラセボ群	実薬群
パターンA	○	○	○	○	○	88	83
パターンB	○	○	○	○	×	6	3
パターンC	○	○	○	×	×	6	6
パターンD	○	○	×	×	×	6	8
パターンE	○	×	×	×	×	14	20




R: 最終評価時点

完全データのベールラインからの変化量 LOCF, Complete case, MAR, pMI, CCMV, NCMV



点線: 観測データ、実線: 完全データ(最初の補完)

完全データのベールラインからの変化量 Delta-adjustment method ($\delta=0, 0.2, 0.4, 0.6, 0.8$)



点線: 観測データ、実線: 完全データ(最初の補完)

補完値の平均値の要約統計量 MI Regression

プラセボ群

	平均	標準偏差	中央値	最小値	最大値
X1の補完値 (N=14)	17.112	0.867	17.130	15.523	19.720
X2の補完値 (N=20)	15.068	1.183	15.170	12.351	17.275
X3の補完値 (N=26)	13.899	1.227	14.064	9.813	16.073
X4の補完値 (N=32)	13.298	1.428	13.005	10.796	16.497

実薬群

	平均	標準偏差	中央値	最小値	最大値
X1の補完値 (N=20)	15.754	0.904	15.678	13.448	17.669
X2の補完値 (N=28)	13.259	1.092	13.230	11.327	15.520
X3の補完値 (N=34)	12.234	1.209	12.014	9.698	14.572
X4の補完値 (N=37)	10.181	1.596	10.197	6.874	13.677

N: その時点の観測値が欠測している症例数

補完値の平均値の要約統計量 Placebo MI

プラセボ群

	平均	標準偏差	中央値	最小値	最大値
X1の補完値 (N=14)	17.090	1.051	16.957	14.781	20.054
X2の補完値 (N=20)	14.718	1.113	14.607	12.219	17.051
X3の補完値 (N=26)	13.659	1.425	13.441	9.654	16.555
X4の補完値 (N=32)	12.908	1.624	13.002	9.782	16.534

実薬群

	平均	標準偏差	中央値	最小値	最大値
X1の補完値 (N=20)	16.594	0.758	16.736	14.278	17.998
X2の補完値 (N=28)	13.392	0.842	13.352	11.618	15.831
X3の補完値 (N=34)	13.294	1.058	13.316	10.431	15.116
X4の補完値 (N=37)	12.427	1.223	12.556	9.591	15.017

N: その時点の観測値が欠測している症例数

補完値の平均値の要約統計量 CCMV

プラセボ群

	平均	標準偏差	中央値	最小値	最大値
X1の補完値 (N=14)	17.117	0.794	17.168	15.480	19.766
X2の補完値 (N=20)	14.987	1.262	15.038	12.248	18.009
X3の補完値 (N=26)	13.770	0.899	13.847	11.657	15.638
X4の補完値 (N=32)	13.203	1.444	12.893	10.631	16.297

実薬群

	平均	標準偏差	中央値	最小値	最大値
X1の補完値 (N=20)	15.359	0.865	15.244	12.979	17.357
X2の補完値 (N=28)	12.871	1.081	12.818	10.578	15.102
X3の補完値 (N=34)	12.040	1.223	12.060	9.499	16.099
X4の補完値 (N=37)	9.862	1.518	9.868	6.671	13.411

N: その時点の観測値が欠測している症例数

補完値の平均値の要約統計量 NCMV(K=1)

プラセボ群

	平均	標準偏差	中央値	最小値	最大値
X1の補完値 (N=14)	17.221	1.384	17.174	14.159	21.369
X2の補完値 (N=20)	18.468	1.381	18.164	15.080	22.283
X3の補完値 (N=26)	19.830	6.139	20.116	5.710	38.088
X4の補完値 (N=32)	17.169	3.646	17.197	8.750	25.296

実薬群

	平均	標準偏差	中央値	最小値	最大値
X1の補完値 (N=20)	16.986	1.144	16.922	14.208	19.403
X2の補完値 (N=28)	15.236	1.615	15.125	12.165	19.346
X3の補完値 (N=34)	10.409	7.375	10.418	-7.099	26.617
X4の補完値 (N=37)	9.713	5.066	10.437	-5.224	20.721

N: その時点の観測値が欠測している症例数

補完値の平均値の要約統計量 NCMV(K=2)

プラセボ群

	平均	標準偏差	中央値	最小値	最大値
X1の補完値 (N=14)	16.529	1.118	16.637	13.833	19.662
X2の補完値 (N=20)	15.308	1.602	15.322	10.340	18.227
X3の補完値 (N=26)	14.132	1.143	14.246	11.529	15.965
X4の補完値 (N=32)	13.437	1.422	13.126	10.452	16.508

実薬群

	平均	標準偏差	中央値	最小値	最大値
X1の補完値 (N=20)	18.289	0.972	18.254	16.168	19.981
X2の補完値 (N=28)	17.321	2.021	16.982	13.649	21.736
X3の補完値 (N=34)	14.668	1.404	14.835	11.696	17.048
X4の補完値 (N=37)	13.054	1.958	12.905	8.694	17.263

N: その時点の観測値が欠測している症例数

補完値の平均値の要約統計量 NCMV(K=3)

プラセボ群

	平均	標準偏差	中央値	最小値	最大値
X1の補完値 (N=14)	13.282	1.328	13.058	10.959	16.239
X2の補完値 (N=20)	13.941	0.926	14.053	12.114	15.611
X3の補完値 (N=26)	14.948	1.220	14.910	11.466	17.986
X4の補完値 (N=32)	16.923	0.896	16.998	14.540	19.722

実薬群

	平均	標準偏差	中央値	最小値	最大値
X1の補完値 (N=20)	17.810	0.928	17.710	15.868	20.612
X2の補完値 (N=28)	14.454	0.961	14.561	12.435	16.979
X3の補完値 (N=34)	13.270	1.225	13.314	10.620	17.279
X4の補完値 (N=37)	11.122	1.475	11.072	7.664	14.728

N: その時点の観測値が欠測している症例数

補完値の平均値の要約統計量 NCMV (K=4)

プラセボ群

	平均	標準偏差	中央値	最小値	最大値
X1の補完値 (N=14)	17.112	0.867	17.130	15.523	19.720
X2の補完値 (N=20)	15.068	1.183	15.170	12.351	17.275
X3の補完値 (N=26)	13.899	1.227	14.064	9.813	16.073
X4の補完値 (N=32)	13.298	1.428	13.005	10.796	16.497

実薬群

	平均	標準偏差	中央値	最小値	最大値
X1の補完値 (N=20)	15.754	0.904	15.678	13.448	17.669
X2の補完値 (N=28)	13.259	1.092	13.230	11.327	15.520
X3の補完値 (N=34)	12.234	1.209	12.014	9.698	14.572
X4の補完値 (N=37)	10.181	1.596	10.197	6.874	13.677

N: その時点の観測値が欠測している症例数

補完値の平均値の要約統計量

Delta-adjustment method (All; $\delta=0.8$)

プラセボ群

	平均	標準偏差	中央値	最小値	最大値
X1の補完値 (N=14)	17.112	0.867	17.130	15.523	19.720
X2の補完値 (N=20)	15.068	1.183	15.170	12.351	17.275
X3の補完値 (N=26)	13.899	1.227	14.064	9.813	16.073
X4の補完値 (N=32)	13.298	1.428	13.005	10.796	16.497

実薬群

	平均	標準偏差	中央値	最小値	最大値
X1の補完値 (N=20)	16.554	0.904	16.478	14.248	18.469
X2の補完値 (N=28)	14.493	1.100	14.494	12.568	16.729
X3の補完値 (N=34)	13.823	1.234	13.640	11.278	16.252
X4の補完値 (N=37)	12.361	1.639	12.390	8.975	15.948

N: その時点の観測値が欠測している症例数

補完値の平均値の要約統計量

Delta-adjustment method (First; $\delta=2.0$)

プラセボ群

	平均	標準偏差	中央値	最小値	最大値
X1の補完値 (N=14)	17.112	0.867	17.130	15.523	19.720
X2の補完値 (N=20)	15.068	1.183	15.170	12.351	17.275
X3の補完値 (N=26)	13.899	1.227	14.064	9.813	16.073
X4の補完値 (N=32)	13.298	1.428	13.005	10.796	16.497

実薬群

	平均	標準偏差	中央値	最小値	最大値
X1の補完値 (N=20)	17.754	0.904	17.678	15.448	19.669
X2の補完値 (N=28)	14.917	1.117	14.901	13.001	17.113
X3の補完値 (N=34)	13.887	1.262	13.790	11.266	16.263
X4の補完値 (N=37)	11.915	1.693	11.932	8.371	15.551

N: その時点の観測値が欠測している症例数

NCMVの場合のPMMの解析結果

補完回数及び1群の症例数による結果の影響

補完回数を変化させた場合の解析結果(MMRM)【症例数:120/群】

補完回数	群間差	群間差のSE	95% CI	P-value
50	-4.009	2.436	-8.848 to 0.830	0.1033
100	-3.987	2.523	-8.964 to 0.990	0.1158
150	-4.139	2.671	-9.395 to 1.118	0.1223
200	-4.194	2.566	-9.239 to 0.850	0.1029

症例数を変化させた場合の解析結果(MMRM)【補完回数:50回】

症例数(/群)	群間差	群間差のSE	95% CI	P-value
120	-4.009	2.436	-8.848 to 0.830	0.1033
150	-3.318	1.562	-6.409 to -0.023	0.0356
180	-2.682	1.398	-5.453 to 0.089	0.0577
200	-2.684	1.122	-4.901 to -0.467	0.0180

NCMVの場合のPMMの解析結果 NCMV (K=2)とNCMV (K=3)の結果【補完回数:50回】

NCMV MMRM	群間差	群間差のSE	95% CI	P-value
NCMV (K=1)	-4.009	2.436	-8.848 to 0.830	0.1033
NCMV (K=2)	-1.990	1.280	-4.506 to 0.526	0.1209
NCMV (K=3)	-2.536	1.167	-4.825 to -0.247	0.0299
NCMV (K=4)	-2.845	1.182	-5.165 to -0.524	0.0163